

## 1. EXAMPLES

```
data F a = Leaf
         | Node a a
deriving Show
```

$$Fa = 1 + a^2$$

```
instance Functor F where
  fmap _ Leaf      = Leaf
  fmap f (Node x y) = Node (f x) (f y)
```

Let's consider, for instance, terms of type `F Int`. It's either a `Leaf` or a `Node` with two numbers in it

```
x1, y1 :: F Int
x1 = Leaf
y1 = Node 1 2
```

Here are some values of the type `F (F Int)`

```
x2, y2 :: F (F Int)
x2 = Leaf
y2 = Node x1 y1
```

We can display `y2`

```
> Node Leaf (Node 1 2)
```

Lifting a function  $f: a \rightarrow b$  to  $F^2$  can be described as applying `fmap` twice. Here's the action of `(+1)` on our test values

```
fmap (fmap (+1)) x2
> Leaf
fmap (fmap (+1)) y2
> Node Leaf (Node 2 3)
```

You can see that `Leafs` at any level remain untouched; only the contents of terminal `Nodes` are transformed.

**1.1. Initial algebra as a colimit.** Now let's apply powers of  $F$  to the initial object, which we define as

```
data Void
deriving Show
```

The `Show` instance for `Void` requires the pragma

```
{-# language EmptyDataDeriving #-}
```

There is only one value of the type `F Void`

```
z1 :: F Void
z1 = Leaf
```

This is a very important property of our  $F$ : its action on the empty set does not produce an empty set. This is what allows us to generate a non-trivial sequence of powers of  $F$  starting with the empty set. The colimit of this sequence is the initial algebra. Au contraire, the same construction for a functor that generates infinite streams

```
data StreamF a x = ConsF a x
```

will produce an uninhabited initial algebra—which is different from its terminal coalgebra, that is

```
data Stream a = ConsF a (Stream a)
```

Double application of our  $F$  to  $\mathbf{Void}$  produces a new  $\mathbf{Leaf}$  and a  $\mathbf{Node}$  that contains two  $\mathbf{Leaf}$ s. Notice the duplication: a  $\mathbf{Leaf}$  is produced at both (in fact, at all non-zero) powers of  $F$  acting on  $\mathbf{Void}$

```
z2, v2 :: F (F Void)
z2 = Leaf
v2 = Node z1 z1
> Node Leaf Leaf
```

Powers of  $F$  acting on  $\mathbf{Void}$  generate trees with terminal  $\mathbf{Leaf}$ s but no terminal  $\mathbf{Nodes}$ . Since we are starting with  $\mathbf{Void}$ , the only function that we can apply or lift is

```
absurd :: Void -> a
absurd a = case a of {}
```

This definition requires another pragma

```
{-# language EmptyCase #-}
```

Lifting `absurd` doesn't change the shapes of these trees.

```
z1' :: F (F Void)
z1' = fmap absurd z1
> Leaf
```

```
z2', v2' :: F (F (F Void))
z2' = fmap (fmap absurd) z2
> Leaf
v2' = fmap (fmap absurd) v2
> Node Leaf Leaf
```

Powers of  $F$  acting on  $\mathbf{Void}$  will eventually produce any tree. But for any given power, there will be even larger trees that are not generated by it. On the other hand, if we take a sum (a coproduct) of all those powers, we'll get all the trees, but also a lot of duplication. For instance, `z1` is the same tree as `z2`, and so on. We have to have a way of identifying multiply generated trees. This is what a colimit of a chain of powers will accomplish.

A colimit in *Set* is a sum, or a discriminated union, in which we identify all the injections that are connected by morphisms in the cocone base.

$$\begin{array}{ccccccc}
 0 & \xrightarrow{!} & F0 & \xrightarrow{F!} & F^2 0 & \cdots & \dots \\
 \downarrow \iota_0 & & \swarrow \iota_{(F0)} & & \searrow \iota_{(F^2 0)} & & \\
 & & \mu F & & & & 
 \end{array}$$

Here we use the lifted **absurd** (or **!** in the picture above) as the morphisms that connect the powers of  $F$  acting on **Void** (or 0 in the picture).

In particular, **fmap absurd** maps the leaf generated by **F Void** to the leaf generated by **F (F Void)**, and so on. All trees generated by the  $n$ 'th power of  $F$  are injected into the  $n + 1$ 'st power of  $F$  by **absurd** lifted by the  $n$ th power of  $F$ . In the colimit is formed by equivalence classes with respect to these identifications. In particular, there is a class for a tree consisting of a single leaf whose representative can be taken from **F Void** or from **F (F Void)** and so on.

**1.2. Terminal coalgebra as a limit.** Things are different with powers of  $F$  acting on the terminal object: singleton or unit type. We get trees with both terminal **Leafs** and terminal **Nodes** that contain units

```

w1, u1 :: F ()
w1 = Leaf
u1 = Node () ()

```

The square of  $F$  produces, among others,

```

w2, u2, t2, s2 :: F (F ())
w2 = Leaf
> Leaf
u2 = Node w1 w1
> Node Leaf Leaf
t2 = Node w1 u1
> Node Leaf (Node () ())
s2 = Node u1 u1
> Node (Node () ()) (Node () ())

```

This time we are interested in mapping into the terminal object

```

unit :: a -> ()
unit _ = ()

```

We are going to lift **unit** to transform  $F^{n+1}1$  to  $F^n 1$ .

Applying **unit** directly to **F ()** turns it into **()**.

Values of the type  $F^2 1$  are mapped to values of the type  $F 1$

```

w2' = fmap unit w2
> Leaf
u2' = fmap unit u2
> Node () ()
t2' = fmap unit t2

```

```

> Node () ()
s2' = fmap unit s2
> Node () ()

```

and so on.

The following pattern emerges.  $F^n 1$  contains trees that end with either leaves or the values of the unit type. The latter play the role of potential seeds. Each seed can give rise to any value of the type  $F ()$ , be it a leaf or a node. Normally, we use a particular coalgebra to generate these values from a carrier, which serves as the seed. Here, we are modeling a situation where a singleton seed generates the whole set of all possible values. It's a "universal" seed.

Except that we are reversing the process: by lifting `unit` we transform  $F^{n+1}$  to  $F^n$ . Instead of expanding the seeds, we are shrinking single-level trees, or buds, back to their seeds. For instance, take

```

t2 :: F (F ())
t2 = Node w1 u1
> Node Leaf (Node () ())

```

which has two buds, `Leaf` and `Node () ()`. `fmap unit` shrinks it down to `Node () ()`. We can think of the "inverse" of the lifted `unit` to generate all possible trees of the next level from the universal seeds embedded in the previous level.

Since a terminal coalgebra is dual to initial algebra, we can generate it as a limit of a sequence that contains powers of  $F$  acting on  $()$  connected by lifted `units`. A limit in *Set* is equivalent to a set of cones whose tip is the singleton set. The base of such cone is a sequence of  $F^n 1$ , starting with 1. The arrows in the base are lifted `units` (or  $j$  in the picture below) pointing from  $F^{n+1} 1$  to  $F^n 1$ .

$$\begin{array}{c}
 1 \\
 \begin{array}{ccc}
 \swarrow \pi_1 & \searrow \pi_{(F1)} & \searrow \pi_{(F^2 1)} \\
 1 & \xleftarrow{j} F1 & \xleftarrow{Fj} F^2 1 \xleftarrow{\dots} \dots
 \end{array}
 \end{array}$$

This chain is infinite, so it's not immediately obvious what part of the cone is relevant. But imagine that the chain terminates on the right. Then the rightmost projection uniquely determines all the others (all the triangles commute).

Notice also that if we select a leaf-terminated tree, such as `Node Leaf Leaf`, then all the values to the right of it (with higher powers of  $F$ ) in the sequence must all be the same. For instance, the third power of  $F$

```

w3 :: F (F (F ()))
w3 = Node Leaf Leaf

```

is mapped to the second power of  $F$  that gives the same tree

```

w3' :: F (F ())
w3' = fmap (fmap unit) w3
> Node Leaf Leaf

```

This way, all finite trees are generated from cones that eventually reach a stable point on the right. However, there are cones that never stabilize, and those correspond to infinite trees.

The **StreamF** functor, in particular, doesn't produce leaves, so all its cones are infinite, corresponding to infinite **Streams**.

## 2. EXISTENCE

$$C^T(\Phi x, a) \cong C(x, Ua)$$

$$C^T(\Phi 0, a) \cong C(0, Ua)$$

$$(Ran_U Id)c = \int_{a \in C^T} a^{C(c, Ua)}$$

$$\mu T = \int_a a^{C(Ta, a)}$$

$$C_T(c, \Psi x) \cong C(Uc, x)$$

$$C_T(c, \Psi 1) \cong C(Uc, 1)$$

$$(Lan_U Id)c = \int^{a \in C_T} C(Ua, c) \cdot a$$

$$\nu T = \int^a C(a, Ta) \cdot a$$

## 3. FIXED POINT

In Haskell, the least fixed point and the greatest fixed point, are often given by the same formula

```
newtype Fix f = Fix { unFix :: f (Fix f) }
```

We can still define them separately by directly encoding their universal property.

## 4. INITIAL ALGEBRA AND CATAMORPHISM

The initial algebra can be defined by its mapping out property.

```
newtype Mu f = Mu (forall a. (f a -> a) -> a)
```

Notice that this definition requires the following language pragma

```
{-# language RankNTypes #-}
```

This definition works because for every least fixed point one can define a catamorphism, which can be rewritten as

```
cata :: Functor f => Fix f -> (forall a . (f a -> a) -> a)
cata (Fix x) = \alg -> alg (fmap (flip cata alg) x)
```

(`flip` is a function that reverses the order of arguments of its (function) argument.) What the definition of `Mu` is saying is that it's an object that, for all algebras, has a mapping out to a catamorphism.

It's easy to define a catamorphism in terms of `Mu`, since `Mu` *is* a catamorphism

```
cataMu :: Functor f => Algebra f a -> Mu f -> a
cataMu alg (Mu cata) = cata alg
```

The challenge is to construct terms of type `Mu f`. For instance, let's convert a list of `a` to a term of type `Mu (ListF a)`

```
mkList :: forall a. [a] -> Mu (ListF a)
mkList as = Mu cata
  where cata :: forall x. (ListF a x -> x) -> x
        cata unf = go as
          where
            go [] = unf NilF
            go (n: ns) = unf (ConsF n (go ns))
```

Notice that we use the type `a` defined in the type signature of `mkList` to define the type signature of the helper function `cata`. For the compiler to identify the two, we have to use the pragma

```
{-# language ScopedTypeVariables #-}
```

You can now verify that

```
cataMu sumAlg (mkList [1..10])
```

produces the correct result for the following algebra

```
sumAlg :: Algebra (ListF Int) Int
sumAlg NilF = 0
sumAlg (ConsF a x) = a + x
```

## 5. TERMINAL COALGEBRA AND ANAMORPHISM

The terminal coalgebra, on the other hand, is defined by its mapping in property. This requires a definition in terms of existential types. If Haskell had an existential quantifier, we could write the following definition for the terminal coalgebra

```
data Nu f = Nu (exists a. (a -> f a, a))
```

Existential types can be encoded in Haskell using the so called Generalized Algebraic Data Types or GADTs

```
data Nu f where
  Nu :: (a -> f a) -> a -> Nu f
```

The use of GADTs requires the language pragma

```
{-# language GADTs #-}
```

The argument is that, for every greatest fixed point one can define an anamorphism

```
ana :: Functor f => forall a. (a -> f a) -> a -> Fix f
ana coa x = Fix (fmap (ana coa) (coa x))
```

We can uncurry it

```
ana :: Functor f => forall a. (a -> f a, a) -> Fix f
ana (coa, x) = Fix (fmap (curry ana coa) (coa x))
```

A universally quantified mapping out

```
forall a. ((a -> f a, a) -> Fix f)
```

is equivalent to a mapping out of an existential type (in pseudo-Haskell)

```
(exists a. (a -> f a, a)) -> Fix f
```

which is the type signature of the constructor of **Nu** *f*.

The intuition is that, if you want to implement a function from an existential type—a type which hides some other type *a* to which you have no access—your function has to be prepared to handle any *a*. In other words, it has to be polymorphic in *a*.

Since in an existential type we have no access to the hidden type, it has to provide both the “producer” and the “consumer” for this type. Here we are given a value of type *a* on the produces side, and the function *a* -> *f a* as the consumer. All we can do is to apply this function to *a* and obtain the term of the type *f a*. Since *f* is a functor, we can lift our function and apply it again, to get something of the type *f (f a)*. Continuing this process, we can obtain arbitrary powers of *f* acting on *a*. We get a recursive data type.

An anamorphism in terms of **Nu** is given by

```
anaNu :: Functor f => Coalgebra f a -> a -> Nu f
anaNu coa a = Nu coa a
```

Notice however that we cannot directly pass the result of **anaNu** to **cataMu** because we are no longer guaranteed that the initial algebra is the same as the terminal coalgebra for a given functor.

## 6. END/COEND FORMULATION

Let’s rewrite **Mu** using GADTs

```
data Mu f where
  Mu :: (forall a. (f a -> a) -> a) -> Mu f
```

We use a natural transformation to construct a **Mu**. Categorically, we can write it as an end

$$\mu f = \int_a a^{C(fa, a)}$$

It’s an end over the profunctor

$$pab = b^{C(fb, a)}$$

Where the power is defined as

$$C(x, a^s) \cong \text{Set}(s, C(x, a))$$

Projection from the end is a catamorphism

$$\pi_a \colon \mu f \rightarrow a^{C(fa, a)}$$

It's a morphism from the hom-set

$$C(\mu f, a^{C(fa, a)})$$

or an element of

$$Set(C(fa, a), C(\mu f, a))$$

Similarly, we can rewrite **Nu**

```
data Nu f where
  Nu :: (a -> f a) -> a -> Nu f
```

as a coend

$$\nu f = \int^a C(a, fa) \cdot a$$

over the profunctor

$$pab = C(a, fb) \cdot b$$

where the copower is defined as

$$C(s \cdot a, x) \cong Set(s, C(a, x))$$

Injection into the coend is an anamorphism

$$\iota_a \colon C(a, fa) \cdot a \rightarrow \nu f$$

It's a morphism from the hom-set

$$C(C(a, fa) \cdot a, \nu f)$$

or an element of

$$Set(C(a, fa), C(a, \nu f))$$

Because of Lambek's lemma, an initial algebra is also a coalgebra, and a terminal coalgebra is also an algebra. Universality, therefore, tells us that there is a unique algebra morphism (as well as a unique coalgebra morphism)

$$\phi \colon \mu f \rightarrow \nu f$$

This is a canonical embedding, but not necessarily an isomorphism.

The two profunctors in the definition of **Mu** and **Nu** can be written as

```
data M f a b = M ((f b -> a) -> b)
```

```
instance Functor f => Profunctor (M f) where
  dimap g g' (M h) = M (\j -> g' (h (g . j . fmap g')))
```

```
data N f a b = N (a -> f b) b
```

```
instance Functor f => Profunctor (N f) where
  dimap g g' (N h b) = N (fmap g' . h . g) (g' b)
```



## 7. HYLOMORPHISM

If the mapping from the terminal coalgebra  $\nu$  to the initial algebra  $\mu$  exists, it is an element of the following hom-set

$$C\left(\int_a^a C(a, fa) \cdot a, \int_b^b b^{C(fb, b)}\right)$$

By co-continuity of the hom-functor, this is isomorphic to

$$\int_a^a C\left(C(a, fa) \cdot a, \int_b^b b^{C(fb, b)}\right)$$

Using continuity we get

$$\int_{a,b} C\left(C(a, fa) \cdot a, b^{C(fb, b)}\right)$$

Using the definition of the copower

$$C(s \cdot a, x) \cong \text{Set}(s, C(a, x))$$

we get

$$\int_{a,b} \text{Set}\left(C(a, fa), C(a, b^{C(fb, b)})\right)$$

And using the definition of the power

$$C(x, a^s) \cong \text{Set}(s, C(x, a))$$

we get

$$\int_{a,b} \text{Set}\left(C(a, fa), \text{Set}(C(fb, b), C(a, b))\right)$$

Finally, applying the currying adjunction we get

$$\int_{a,b} \text{Set}\left(C(a, fa) \times C(fb, b), C(a, b)\right)$$

in which you may recognize a hylomorphism

```
hylo :: Functor f => Coalgebra f a -> Algebra f b -> a -> b
hylo coa alg = alg . fmap (hylo coa alg) . coa
```

## 8. BIBLIOGRAPHY

- Michael Barr, [Terminal coalgebras for endofunctors on sets](#)

## 9. RANDOM THINGS

### 9.1. Initial algebra structure map.

$$j: f(\mu f) \rightarrow \mu f$$

$$C\left(f(\mu f), \int_b^b b^{C(fb, b)}\right)$$

is a member of

$$\int_b^b C\left(f(\mu f), b^{C(fb, b)}\right)$$

Using the definition of the power

$$C(x, a^s) \cong \text{Set}(s, C(x, a))$$

we get

$$\int_b \text{Set}(C(fb, b), C(f(\mu f), b))$$

Using Yoneda lemma we replace  $b$  with  $f(\mu f)$

$$C(f(f(\mu f)), f(\mu f))$$

## 9.2. Terminal coalgebra structure map.

$$k: \nu f \rightarrow f(\nu f)$$

is a member of

$$C\left(\int^a C(a, fa) \cdot a, f(\nu f)\right)$$

$$\int_a C\left(C(a, fa) \cdot a, f(\nu f)\right)$$

Using the definition of the copower

$$C(s \cdot a, x) \cong \text{Set}(s, C(a, x))$$

we get

$$\int_a \text{Set}\left(C(a, fa), C(a, f(\nu f))\right)$$

Yoneda lemma

$$C(f(\nu f), f(f(\nu f)))$$

## 9.3. Kan extensions.

$$\mu f = \int_a a^{C(fa, a)}$$

$$\nu f = \int^a C(a, fa) \cdot a$$

These formulas are reminiscent of Kan extensions. For comparison, the right Kan extension of  $g$  along  $f$  is given by

$$(Ran_f g)c = \int_a (ga)^{C(c, fa)}$$

The left Kan extension is

$$(Lan_f g)c = \int^a C(fa, c) \cdot ga$$

If  $f$  has left and right adjoints, they are given by

$$Ran_f Id \dashv f \dashv Lan_f Id$$

In particular, using the adjunction

$$(Lan_f Id)c = \int^a C(a, (Lan_f Id)c) \cdot a$$

This shows that  $(Lan_f Id)c$  is a fixed point of the functor

$$\Phi(x) = \int^a C(a, x) \cdot a$$

**9.4. Ends as limits.** Twisted arrow category on  $Tw(\mathbf{C})$  has, as objects, morphisms in  $\mathbf{C}$  (or, strictly speaking, triples  $(a, b, f: a \rightarrow b)$ ). A morphism from  $f: a \rightarrow b$  to  $g: a' \rightarrow b'$  is a pair of morphisms

$$(h: a' \rightarrow a, h': b \rightarrow b')$$

For every profunctor  $p: C^{op} \times C \rightarrow \mathbf{Set}$  define a functor  $\bar{p}: Tw(\mathbf{C}) \rightarrow \mathbf{Set}$ . On objects

$$\bar{p}(a, b, f) = pab$$

and on morphisms, it's just profunctor lifting.

It can be shown that the end is just a limit over the twisted arrow category

$$\int_c pcc \cong \lim_{Tw(C)} \bar{p}$$

Similarly, the coend is a colimit over  $Tw(C^{op})^{op}$

$$\int^c pcc \cong \operatorname{colim}_{Tw(C^{op})^{op}} \bar{p}$$

**9.5. Iterative solution.** Terminal coalgebra is a limit, and initial algebra is a colimit of these two chains

$$\begin{array}{c}
 f(\nu f) \\
 \uparrow k \\
 \nu f \xrightarrow{\pi_1} 1 \xleftarrow{i} f1 \xleftarrow{f!} f^2 1 \xleftarrow{\dots} \dots \\
 \downarrow \pi_1 \quad \searrow \pi_{(f1)} \quad \quad \quad \searrow \pi_{(f^2 1)} \\
 0 \xrightarrow{!} f0 \xrightarrow{f!} f^2 0 \xrightarrow{\dots} \dots \\
 \downarrow \iota_0 \quad \swarrow \iota_{(f0)} \quad \quad \quad \swarrow \iota_{(f^2 0)} \\
 \mu f \xleftarrow{j} f(\mu f)
 \end{array}$$

$\phi$  (curved arrow from  $\mu f$  to  $\nu f$ )